

Appendix Text 3. WinBUGS code for Model 3.

```

model {
  for (i in 1:N) {
    y[i] ~ dpois(mu[i])
    log(mu[i]) <- log(n[i]) + epsilon[age[i],LAD[i],yr[i]]
  }

#####
##### priors on the overall intercept (alpha) and overall slope (beta)
#####
alpha ~ dnorm(0, 0.00001)
beta ~ dnorm(0, 0.00001)

#####
##### age and district specific intercepts and age-district interactions
#####

#####
##### age intercepts with RW1 prior
theta[1:Ages] ~ car.normal(adj.age[],weights.age[],num.age[],tau_theta)
sigma_theta ~ dunif(0,2)
tau_theta <- pow(sigma_theta,-2)

#####
##### district intercepts with BYM prior
u[1:LADs] ~ car.normal(adj.LAD[],weights.LAD[],num.LAD[],tau_u)
for (d in 1:LADs) {
  tmp[d] <- alpha + u[d]
  v[d] ~ dnorm(tmp[d],tau_v)
}
sigma_u ~ dunif(0,2)
tau_u <- pow(sigma_u,-2)
sigma_v ~ dunif(0,2)
tau_v <- pow(sigma_v,-2)

#####
##### age-district interactions
for (a in 1:Ages) {
  for (d in 1:LADs) {
    mu.age_LAD[a,d] <- theta[a] + v[d]
    age_LAD[a,d] ~ dnorm(mu.age_LAD[a,d],tau_age_LAD)
  }
}
sigma_age_LAD ~ dunif(0,2)
tau_age_LAD <- pow(sigma_age_LAD,-2)

#####
##### age/cohort related time trends
#####
for (a in 1:Ages) {
  gamma.temp[a,1] <- 0
  gamma[a,1] <- 0
  beta_age_all[a] <- beta + beta_age[a]
  for (tt in 2:obs.Years) {
    gamma.temp[a,tt] <- gamma[a,tt-1] + beta_age_all[a] + beta_cohort[cohort.ind[a,tt]]
    gamma[a,tt] <- gamma.temp[a,tt] + add.noise[a,tt] * noise[a,tt]
  }
  # Note: add.noise is a matrix with all entries being 1 apart from add.noise[19,1:5]=0 and
  # add.noise[1,(obs.Years-4):obs.Years]=0 to suppress the random walk part for identifiability of
}

```

```

# the oldest and the youngest cohorts

noise[a,tt] ~ dnorm(0,tau_gamma)
}}
sigma_gamma ~ dunif(0,2)
tau_gamma <- pow(sigma_gamma,-2)

#####
#### age slopes with RW1 prior
#####
beta_age[1:Ages] ~ car.normal(adj.age[],weights.age[],num.age[],tau_beta_age)
sigma_beta_age ~ dunif(0,2)
tau_beta_age <- pow(sigma_beta_age,-2)

#####
#### district related trends
#####
for (d in 1:LADs) {
  gamma_LAD.temp[d,1] <- 0
  gamma_LAD[d,1] <- 0
  for (tt in 2:obs.Years) {
    gamma_LAD.temp[d,tt] <- gamma_LAD[d,tt-1] + beta_LAD[d]
    gamma_LAD[d,tt] ~ dnorm(gamma_LAD.temp[d,tt],tau_gamma_LAD)
  }
}
sigma_gamma_LAD ~ dunif(0,2)
tau_gamma_LAD <- pow(sigma_gamma_LAD,-2)

#####
#### district slopes with BYM prior
#####
beta_LAD_u[1:LADs] ~ car.normal(adj.LAD[],weights.LAD[],num.LAD[],tau_beta_LAD_u)
for(d in 1:LADs) {
  beta_LAD.temp[d] ~ dnorm(beta_LAD_u[d],tau_beta_LAD)
  beta_LAD[d] <- beta_LAD.temp[d]
}
sigma_beta_LAD_u ~ dunif(0,2)
tau_beta_LAD_u <- pow(sigma_beta_LAD_u,-2)
sigma_beta_LAD ~ dunif(0,2)
tau_beta_LAD <- pow(sigma_beta_LAD,-2)

#####
#### putting all together
#####
for (a in 1:Ages) {
  for (d in 1:LADs) {
    epsilon[a,d,1] <- age_LAD[a,d]
    for (tt in 2:obs.Years) {
      mu.epsilon[a,d,tt] <- age_LAD[a,d] + gamma[a,tt] + gamma_LAD[d,tt]
      epsilon[a,d,tt] ~ dnorm(mu.epsilon[a,d,tt],tau_epsilon)
    }
  }
}
sigma_epsilon ~ dunif(0,2)
tau_epsilon <- pow(sigma_epsilon,-2)
}

```